# Maintainer: Felix Yan <felixonmars@archlinux.org>

pkgname=fingerterm
pkgver=1.4.7
_commit=cfaeaa3bf91f12fa96eaa78f0dd7dbc94375fabb
pkgrel=1
pkgdesc="A terminal emulator with a custom virtual keyboard"
arch=('loong64' 'x86_64')
url="https://github.com/sailfishos/fingerterm"
license=('LGPL')
depends=('qt5-feedback')
makedepends=('git' 'nemo-qml-plugin-notifications' 'qt5-tools')
source=("git+https://github.com/sailfishos/fingerterm.git#commit=$_commit")
sha512sums=('SKIP')

build() {
  cd fingerterm

  qmake-qt5 CONFIG+=enable-feedback CONFIG+=enable-nemonotifications
  make
}

package() {
  cd fingerterm

  make INSTALL_ROOT="$pkgdir" install
}
